

New Multi Phase Quasi Static Fundamental Wave Electric Machine Models for High Performance Simulations

Christian Kral Anton Haumer
Electric Machines, Drives and Systems Technical Consulting
1060 Vienna, Austria 3423 St.Andrä-Wördern, Austria
dr.christian.kral@gmail.com a.haumer@haumer.at

Abstract

A new quasi static fundamental wave machines library will be included in the magnetic domain package of the next Modelica Standard Library (MSL). The provided classes of machine models omit all transient electrical effects, but mechanical dynamics are fully taken into account. By including the new machine models new classes of problems can be treated, enabling fast electric machine and drive simulations. Yet, all the characteristic loss effects of transient machine models are fully taken into account, where needed. Phase numbers greater than three are supported. For each machine type available in the MSL there will then exist both a fully transient and a quasi static electric machine model. The package structure of the quasi static fundamental wave package and the concept of implementation will be presented. All required assumptions and limitations for operating the new machine models will be presented and discussed. Deviating parameters compared to the transient machine models will be discussed and explained. Simulation examples will be presented and compared with transient simulation experiments. Possible applications for the new machine models will be outlined.

Keywords: Quasi static fundamental wave electric machine models, multi phase, transient effects, reference frame

1 Introduction

The investigation of electric machine transients is highly relevant for designing drive controls and investigating complex multi domain physical systems. Yet, in many applications the full electrical transients play a minor role. For example, the investigation of electric driving cycles of full or hybrid electric vehicles or railways or the assessment of auxiliary drives or the superior control of multiple drives in an industrial environment do not require the full consideration of all electrical transients. In such cases it is assumed that machine control is designed and implemented such way that electric transients have no significant impact either on power and energy balance or on the overloading capability or stability. Additionally, it may even be requested or required to reduce system complexity for such applications

by eliminating electrically transient phenomena. The newly developed quasi static fundamental wave electric machines library is closing the gap for fast, simple and accurate electric machine models in a Modelica simulation environment. The new quasi static fundamental wave electric machines library is designed in the style of the transient FundamentalWave library [1] which is already included in the Modelica Standard Library (MSL). The new library utilizes the existing quasi static multi phase electric package as well as the rotational mechanics and the thermal domain [2]. All machine models are strictly object oriented: Modelica classes for windings, air gaps, linear magnetic reluctances, losses, etc. are provided. From a didactic point of view the quasi static magnetic field theory enables valuable insight into electro magnetic and mechanic power conversion. Additionally, the provided linear machine models are designed such way that they can be extended towards non linear magnetic effects such as saturation or permanent magnet (PM) demagnetization. When designing the new library it was intended to include as much features as provided by the transient machine models. Therefore, phase numbers greater than three and different phase numbers in the stator and the slip ring rotor of induction machines are fully supported. Yet the number of phases, m , being equal to 2^n with integer n are currently not included. The reason for omitting such phase numbers is caused by the fundamental difference of windings systems with $m = 2^n$ from other windings.

An alternative concept for modeling quasi static magnetic machines in Modelica is presented in [3]. In this concept an induction machine model is presented based solely on electric and magnetic quasi static equivalent circuits. Electro mechanical power conversion is not yet included in this proposal.

2 Connector Concept

From a formal point of view the connector concept of the transient and the quasi static fundamental wave package look very similar. The magnetic port consists of the complex magnetic potential, $\underline{V}_m = V_{m, \text{re}} + jV_{m, \text{im}}$, and the complex magnet flux, $\underline{\Phi} = \Phi_{\text{re}} + j\Phi_{\text{im}}$.

```
connector MagneticPort
  "Basic quasi static magnet connector"
```




Figure 1: Stator and rotor fixed reference frame of an electric machine

```

Modelica.SIunits.ComplexMagneticPotential
  V_m "Complex magnetic potential
    at the node";
  flow Modelica.SIunits.ComplexMagneticFlux
    Phi "Complex magnetic flux flowing
    into the pin";
end MagneticPort;
    
```

The complex magnetic potential and the complex magnetic flux represent a spatial fundamental wave field distribution given by:

$$\begin{aligned}
 \Phi(\varphi) &= \text{Re}[(\Phi_{\text{re}} + j\Phi_{\text{im}})e^{-j\varphi}] \\
 &= \Phi_{\text{re}} \cos(\varphi) + \Phi_{\text{im}} \sin(\varphi) \\
 V_m(\varphi) &= \text{Re}[(V_{m,\text{re}} + jV_{m,\text{im}})e^{-j\varphi}] \\
 &= V_{m,\text{re}} \cos(\varphi) + V_{m,\text{im}} \sin(\varphi)
 \end{aligned}$$

A visualization of the fundamental wave forms is illustrated in [1].

The main difference is the reference angle included in both the quasi static positive and negative connector class definition.

```

connector PositiveMagneticPort
  "Positive magnetic port"
  extends QuasiStaticFundamentalWave.
    Interfaces.MagneticPort;
  Modelica.Electrical.QuasiStatic.
    Types.Reference reference "Reference";
end PositiveMagneticPort;
    
```

The reference angle represents the angle of the reference frame that the connector refers to. In electric machines typically stator and rotor reference frames are used in which the fundamental principles of Ampere's law and induction law apply. The stator of electric motors are either supplied by a fixed or variable frequency source. Stator frequency determines the stator reference angle. The frequency of the induced voltages of the rotor depends on the machine type and whether the winding is accessible from the outside. In case of a synchronous generator stator frequency is determined by the rotational speed.

Stator and rotor reference frames with respect to a complex magnetic flux phasor are depicted in Fig. 1. The complex phasor consists of a real and imaginary part, in the respective reference frame. The different reference frames are, however, indicated by the reference angle provided by the connector. So when coupling the stator and rotor fixed reference frame over the air gap model the real and imaginary

parts of magnetic flux are the same. Only the reference angles are different. The very same applies for the complex magnetic potential difference.

In Fig. 1 (s) represents the stator fixed reference frame and γ_s is the angle difference between connector reference frame and the stator fixed reference frame. In the same way, (r) represents the rotor fixed reference frame and γ_r is the angle difference between connector reference frame and the rotor fixed reference frame. The angle difference

$$\gamma = \gamma_s - \gamma_r$$

is equivalent to the rotational mechanical angle between stator and rotor, multiplied by the number of pole pairs, p .

3 Assumptions and Limitations

First, in the actual version of the quasi static fundamental wave library the phase number m must not be equal to 2^n for integer n .

Second, the used electric connector is based on the package `Modelica.Electrical.QuasiStatic.MultiPhase`. Therefore, only single frequency voltages and currents are taken into account in the underlying electrical connector concept.

The third limitation is that only symmetric voltages and currents are allowed for supplying and loading the machines. This restriction is a consequence of the fact that unbalanced voltage and current supply, respectively, cause forward and backward spinning magnetic field waves. In the squirrel cages of induction machines the backwards spinning field waves give rise to slip dependent frequency components in the voltages and currents which cannot be taken into account by the single supply frequency approach of the quasi static multi phase connector. When investigating the power flow and systemic behaviors of mains supplied electric machines any supply asymmetrical is usually of minor interest. For variable speed inverter fed electric machines symmetrical voltage supply can be assumed due to the control of the power electronics which strictly avoids voltage asymmetries.

Fourth, all windings are assumed to be fully symmetrical. So the numbers of turns are equal for all winding axes and the winding axis orientations are strictly related with the function shown in Listing 1.

Fifth, due to the assumed symmetry of supply voltages and currents, zero voltages, zero currents and zero impedances are not considered in the quasi static machine models.

Sixth, all the magnetic reluctances of the machine models are assumed to be constant. This represents a strictly linear relationship between magnetic potential differences and magnetic fluxes.

4 Library Structure and Components

The structure of the new quasi static fundamental wave library is presented in Fig. 2. The key components such as




Figure 2: Structure of the quasi static fundamental wave library

the electro magnetic coupling, the salient air gap model and the damper cage concept which will be presented in the following subsections. The coupling models rely on the symmetry of voltages and currents. Therefore, symmetrical components are discussed in this section as well. Cage models, the PM model, regular reluctance and eddy current models are designed in the style of the transient fundamental wave library and need no particular attention in this paper. More detailed model descriptions can be found in [1]. Parametrization rules for multi phase machines with phase numbers greater or equal to three are described in [4].

4.1 Symmetrical Components

The orientations of the winding axes of an m -phase system is defined by the function listed in Listing 1, which is also used in the transient fundamental wave library.

Listing 1: Function symmetricOrientation

```
function symmetricOrientation
  "Orientations of the resulting
  fundamental wave field phasors"
  extends Modelica.Icons.Function;

  input Integer m "Number of phases";
```

```
output Modelica.SIunits.Angle
orientation[m]
"Orientation of the resulting
fundamental wave field phasors";

import Modelica.Constants.pi;

algorithm
  if mod(m, 2) == 0 then
    // Even number of phases
    if m == 2 then
      // Special case two phase machine
      orientation[1] := 0;
      orientation[2] := +pi/2;
    else
      orientation[1:integer(m/2)] :=
        symmetricOrientation(integer(m/2));
      orientation[integer(m/2) + 1:m] :=
        symmetricOrientation(integer(m/2))
        - fill(pi/m, integer(m/2));
    end if;
  else
    // Odd number of phases
    orientation :=
      {(k - 1)*2*pi/m for k in 1:m};
  end if;
end symmetricOrientation;
```

An arbitrary m -phase system of currents $i_{[k]}$, for $1 \leq k \leq m$, can be transformed into m symmetrical components. In case of a fully symmetrical system of currents, only the positive sequence component is non-zero. The positive sequence component is computed by means of multiplying the current vector with the transformation matrix obtained from function `symmetricOrientationMatrix`. The design of the transformation matrix relies on the recursive application of the function presented in Listing 1.

As a consequence of assuming fully symmetrical voltages and currents, respectively, only the positive sequence of voltages and currents arise. In case of even phase numbers more than one positive sequence component will arise. It is thus required to determine the indexes of positive sequence components by means of function `indexPositiveSequence`. All other symmetrical components are equal to zero. Their indexes are determined by function `indexNonPositiveSequence`.

4.2 Electromagnetic Coupling

The induction law and Ampere's law are related with the positive sequences of the symmetrical components of voltages and currents. Due to the symmetry of the windings the induced voltages of all positive sequence components are identical. In the quasi static domain the time derivative of the magnetic flux is replaced by a multiplication with the imaginary unit and the angular frequency. The complex total magnetic potential difference is related with the sum of all positive sequence currents, see Listing 2.

Listing 2: Electromagnetic coupling model incorporating the induction law and Ampere's law

```
model MultiPhaseElectroMagneticConverter
  "Multi phase electro magnetic converter"
  ...
  QuasiStationary.MultiPhase.Interfaces.
```

```

    PositivePlug
    plug_p(final m=m) "Positive plug";
    QuasiStationary.MultiPhase.Interfaces.
    NegativePlug
    plug_n(final m=m) "Negative plug";
    Interfaces.PositiveMagneticPort port_p
    "Positive complex magnetic port";
    Interfaces.NegativeMagneticPort port_n
    "Negative complex magnetic port";
    Modelica.SIunits.ComplexVoltage
    v[m] "Voltage drop";
    Modelica.SIunits.ComplexCurrent
    i[m] "Current";
    ...
    SIunits.ComplexVoltage
    vSymmetricalComponent[m] =
    symmetricTransformationMatrix(m)*v
    "Symmetrical components of voltages";
    SIunits.ComplexCurrent
    iSymmetricalComponent[m] =
    symmetricTransformationMatrix(m)*i
    "Symmetrical components of currents";
protected
    final parameter Integer indexNonPos[:] =
    indexNonPositiveSequence(m)
    "Indices of all non positive sequence
    componentes";
    final parameter Integer indexPos[:] =
    indexPositiveSequence(m)
    "Indices of all positive sequence
    componentes";
equation
    // Magnetic flux and flux balance
    // of the magnetic ports
    port_p.Phi = Phi;
    port_p.Phi + port_n.Phi = Complex(0,0);
    // Magnetic potential difference
    // of the magnetic ports
    port_p.V_m - port_n.V_m = V_m;
    // Voltage drop of electrical plugs
    v = plug_p.pin.v - plug_n.pin.v;
    // Current and current balance of plugs
    i = plug_p.pin.i;
    plug_p.pin.i + plug_n.pin.i =
    {Complex(0,0) for k in 1:m};
    // Amperes law
    V_m.re = sqrt(2) * (2.0/pi) *
    Modelica.ComplexMath.real(
    N*iSymmetricalComponent[1])*m/2;
    V_m.im = sqrt(2) * (2.0/pi) *
    Modelica.ComplexMath.imag(
    N*iSymmetricalComponent[1])*m/2;
    for k in 1:size(indexNonPos,1) loop
    iSymmetricalComponent[indexNonPos[k]] =
    Complex(0,0);
    end for;
    // Induction law
    for k in 2:size(indexPos,1) loop
    vSymmetricalComponent[indexPos[1]] =
    vSymmetricalComponent[indexPos[k]];
    end for;
    -sqrt(2) * Complex(
    Modelica.ComplexMath.real(
    vSymmetricalComponent[indexPos[1]]),
    Modelica.ComplexMath.imag(
    vSymmetricalComponent[indexPos[1]]))
    = Modelica.ComplexMath.conj(N)*j*omega*Phi;
    // Breakable connections of references
    Connections.branch(
    port_p.reference, port_n.reference);
    port_p.reference.gamma =
    port_n.reference.gamma;
    Connections.branch(

```

```

    plug_p.reference, plug_n.reference);
    plug_p.reference.gamma =
    plug_n.reference.gamma;
    Connections.branch(
    plug_p.reference, port_p.reference);
    plug_p.reference.gamma =
    port_p.reference.gamma;
    ...
end MultiPhaseElectroMagneticConverter;

```

The reference angles of both the electrical and the magnetic domain are connected by means of breakable connections. This concept breaks potential algebraic loops of the electric and magnetic domain. Additionally, both domains are interconnected by a breakable connector as well.

4.3 Air Gap

The air gap model consists of two stator and two rotor magnetic ports and two rotational mechanic flanges representing the stator and rotor, respectively. The complex magnetic potential difference and flow quantities of the quasi static fundamental wave stator and rotor connectors are identical. However, the reference angles are different according to Fig. 1.

Rotor saliency is represent by different magnetic reluctances in the d (direct) and q (quadrature) axis – with respect to the rotor fixed reference frame. Therefore, the relationships between magnetic potential differences and fluxes have to be expressed in the rotor fixed reference frame: the rotor fixed complex magnetic potential differences and fluxes are obtained by multiplying each of these quantities by $e^{j\theta_r}$ as shown in Listing 3.

Listing 3: Air gap model with rotor saliency

```

model RotorSaliencyAirGap "Air gap model
with rotor saliency"
    Interfaces.PositiveMagneticPort
    port_sp "Positive complex magnetic
    stator port";
    Interfaces.NegativeMagneticPort
    port_sn "Negative complex magnetic
    stator port";
    Interfaces.PositiveMagneticPort port_rp
    "Positive complex magnetic rotor port";
    Interfaces.NegativeMagneticPort port_rn
    "Negative complex magnetic rotor port";
    Rotational.Interfaces.Flange_a flange_a
    "Flange of the rotor";
    Rotational.Interfaces.Flange_a support
    "Support at which the reaction torque
    is acting";
    parameter ...
    // Phasors of magn. potential differences
    SIunits.ComplexMagneticPotentialDifference
    V_ms
    "Complex magnetic potential difference
    of stator w.r.t. stator ref. frame";
    SIunits.ComplexMagneticPotentialDifference
    V_msr = V_ms *
    ComplexMath.fromPolar(1,gammar)
    "Complex magn. potential difference of
    stator w.r.t. rotor fixed ref. frame";
    ...
    Modelica.SIunits.Torque tauElectrical
    "Electrical torque";
    Modelica.SIunits.Angle gamma =
    p*(flange_a.phi - support.phi)

```

```

    "Electr. angle betw. rotor and stator";
    SIunits.Angle gammas =
      port_sp.reference.gamma
    "Angle in stator reference frame";
    SIunits.Angle gammar =
      port_rp.reference.gamma
    "Angle in rotor reference frame";
equation
...
// Local balance of magneto motive force
// w.r.t. rotor fixed reference frame
(pi/2.0)*(V_mrr.re + V_msr.re) =
  Phi_rr.re*R_m.d;
(pi/2.0)*(V_mrr.im + V_msr.im) =
  Phi_rr.im*R_m.q;
// Torque
tauElectrical = -(pi*p/2.0)*
  (Phi_s.im*V_ms.re - Phi_s.re*V_ms.im);
flange_a.tau = -tauElectrical;
support.tau = tauElectrical;
// Stator may be potential root
Connections.potentialRoot
  (port_sp.reference);
Connections.branch(
  port_sp.reference, port_sn.reference);
port_sp.reference.gamma =
  port_sn.reference.gamma;
Connections.branch(
  port_rp.reference, port_rn.reference);
port_rp.reference.gamma =
  port_rn.reference.gamma;
Connections.branch(
  port_sp.reference, port_rp.reference);
gammas = gammar + gamma;
if Connections.isRoot(port_sp.reference)
then
  gammar=0;
end if;
end RotorSaliencyAirGap;

```

The references of the stator and rotor ports are linked through breakable connections. In case of generator operation without mains connection the stator will have to be treated as root.

4.4 Damper Cages

For operating mains supplied synchronous machines it is required to enable the optional damper cage. Otherwise the induced rotor voltage for electrical and PM excited synchronous machines cannot perform the expected slow motion rotation. Since the load angle cannot change abruptly due to rotor inertia, the damper cage is required for operating quasi static synchronous machines with mains supply. The operating behavior of the quasi static machine model is yet different since all transient electrical effects are not taken into account.

The optional damper cages of the synchronous machines are equivalent two axis cages with different resistances and inductances in the rotor d (direct) and q (quadrature) axis. The squirrel cage of the induction machine model, however, has the same number of phases as the stator. Same stator and rotor phase numbers are chosen since it is usual in engineering approaches to model the squirrel cage equivalent to the topology of the stator winding.

4.5 Example Machine Model

Figure 3 shows the permanent magnet synchronous machine model included in the quasi static fundamental wave library. The quasi static electrical domain is colored in light blue, the quasi static fundamental wave domain is light orange. The rotational and the thermal domain are colored in black and red, respectively.

The stray load loss model is directly part of the stator electric circuit. The voltage drop of this model is equal to zero, and loss is considered as torque times angular frequency. The stray load loss is also dissipated through the thermal connector.

The stator winding represents the electro magnetic coupling of the stator winding with magnetic field. The winding model is depicted in Fig. 4. This model consists of symmetrical winding phase resistors, an ideal electromagnetic coupling, a stray field reluctance in the magnetic domain and a loss model representing solely eddy current core loss. Thermal connectors for copper loss and eddy current loss are provided. Winding resistances are modeled temperature dependent, eddy current loss is modeled independent of temperature.

The air gap model couples the stator and rotor magnetic parts of the model. The electromagnetic torque is transmitted to the stator and rotor, respectively. Stator and rotor torques have the same absolute value but different signs. The magnetic saliency of the rotor is considered by different stator inductances with respect to the d (direct) and q (quadrature) axis. In the actual implementation the total inductances of the two axes are associated to one reluctance with respect to the d and q axis, respectively, only. Partial reluctances of the stator and rotor teeth, slots, yokes, etc. are currently not separated.

In the rotor magnetic circuit an optional damper cage is included. The damper cage model is a two axis model with cage resistances and stray inductances assigned to the two axis – this is the usual parametrization that electrical engineers are using. The damper cage also provides a thermal connector to exchange heat and temperature with an optional thermal model.

The permanent magnet model is currently not considering temperature dependent magnet properties. The total inductance of the machine already includes the reluctance of the permanent magnet. The PM model is thus a constant source of magnetic potential difference, rotated into the rotor fixed reference frame such that the complex magnetic potential difference phasor is aligned with the d axis. The mechanical flanges of the PM model are required to model the PM loss as equivalent mechanical loss torque times angular velocity.

The inertias, the friction loss model and the heat port and ambient models are the same for the quasi static and the transient fundamental wave machine models.




Figure 3: Quasi static model of a multi phase permanent magnet synchronous machine with permanent magnet

5 Library Compatibility

5.1 Parameter Compatibility

The new quasi static fundamental wave library is almost fully parameter compatible with the transient fundamental wave library. The only incompatibility are the stator zero inductances and the rotor zero inductance of the induction machine with wound slip ring rotor. These zero inductances are not implemented in quasi static machine models. In the transient machine models the default values of the zero inductances are the stray inductances. If the user does not propagate an actual parameter for the zero inductance of a transient machine model, the quasi static and transient machine models can be exchanged without parameter inconsistency.

In the current version of the quasi static fundamental wave library the loss models are fully compatible with the transient machine models. This may change in future versions where quasi static hysteresis loss may be included – which are difficult to implement for transient magnetic fields. Stator core loss, friction loss, stray load loss, permanent magnet loss and brush loss are implemented with compatible parameters and equal static state behavior.

5.2 Number of Phases

Parameter compatibility also includes the number of phases. In the quasi static fundamental wave library multi phase machines with phase numbers greater than or equal to three are supported.




Figure 4: Winding model including winding resistors, electromagnetic coupling, stray reluctances and eddy current loss

5.3 Connector Compatibility

The electrical AC multi phase connectors of the transient and quasi static machine model are not compatible due to the different connector designs and properties. The analog DC connector of the synchronous machine with electrical excitation is yet connector compatible. The thermal connectors and the rotational connectors of the shaft and the optional housing, respectively, are connector compatible with the transient machine models.

6 Examples

In the following examples of transient and quasi static fundamental wave electric machines are compared. Each of the published examples is also available at the sub package `Examples.BasicMachines`.

6.1 Induction Machine with Squirrel Cage Rotor

Starting an induction machine with squirrel cage rotor direct on line shows a transiently higher starting and a lower break down torque than the quasi static machine; see Fig. 5 and 6. The simulation result is not plotted as a function of time, but as parametric plot of torques versus speed. Both machines are loaded with an idealized mechanical load, modeled as quadratic speed dependent torque. The example is available at `Examples.BasicMachines.AIMC_DOL`.




Figure 5: Modelica model of both transient and quasi static squirrel cage induction machine starting directly on line (DOL)

6.2 Electrical Excited Synchronous Generator

In example `Examples.BasicMachines.SMEE_Generator` a transient and quasi static synchronous generator with electrical excitation are compared for a very slow load change. The shafts of both generator are spinning with a rotational speed slightly different than synchronous speed, see Fig. 7. Therefore, the rotor is moved over one full electrical revolution – relative to the magnetic field. In the actual Modelica example the total real time of the experiment is equal to 30s. Such an experiment can also be performed in the lab for determining the full torque characteristic as function of the load angle. In Fig. 8 the torques of both machines are displayed versus angle γ_r obtained by the quasi static machine model. Angle γ_r and the load angle ϑ are related by

$$\vartheta = \gamma_r - 90^\circ.$$

The torques of both machines are identical due to small relative speed between rotor and magnetic field. The wave form the torque is a sine wave superimposed with a sine wave of half the period due to the rotor saliency of this machine.

7 Possible Applications

7.1 Long Periods of Simulation Time

The new quasi static fundamental wave models may be used for all time domain simulations covering a large time span. Neglecting electrical transient effects makes the simulation models fast and robust. In particular, large periods of simulation time being in the range of minutes, hours and days, require fast simulation models. In typical simulation applications it is important to model power and energy balances accurately. High accuracy of the actual efficiency is




Figure 6: Torque versus speed for an induction machine with squirrel cage rotor, starting direct on line; comparing transient (trans) and quasi static (QS) machine model; load torque is quadratic speed dependent on speed




Figure 7: Modelica models of synchronous generators with electrical excitation




Figure 8: Torques of the transient and quasi static generator versus angle γ_r

enabled by comprehensive loss models included in the machine models of the quasi static fundamental wave library. Typical applications are:

- Drive cycles of full or hybrid electric vehicles
- Drive cycles of subway trains, tramways, and railway trains
- Modeling of mechanical and electrical power and energy balances in industrial processes
- Robot drives for industrial use
- Auxiliary drives

7.2 Modularity of Simulation Concepts

Particularly for the simulation of electric machines including power electronics (converter) and control quasi static drive models are an interesting option. Quasi static and transient electric machine models are different levels of abstraction. The two different machine models can be combined with either a quasi static and transient converter models. Both the quasi static and the transient machine and converter can be operated by one control algorithm [5–10]. This way, the level of modularity of electric drives can be increased by means of the new quasi static fundamental wave machine models. The increased modularity reduces maintenance effort for drives libraries and reduces the development time of new or alternative control algorithms, in particular with respect to multi phase electric drives with phase numbers greater than three.

8 Didactic Aspects

In the quasi static and the transient fundamental wave library a strictly object oriented modeling approach was pursued. This includes the physical domains electrical, magnetic, rotational and thermal. Due to this approach the models of windings, the air gap and the permanent magnet, etc., can be clearly separated as different objects. In future versions it is even possible to separate the magnetic reluctances of teeth, slots, yokes and magnets.

Quasi static machine models are also of particular interest for high schools and universities to make virtual experiments and to demonstrate the physical behavior of electric machines. The different behavior and quality of quasi static and transient machine models can be discussed and elaborated in classes.

9 Conclusions

The new quasi static fundamental wave library for modeling induction and synchronous electric machines is presented. This package shall be included in the next version of the MSL. The provided machine models allow phase number equal to or greater than three. Fully symmetrical windings

and supply voltages and currents are required to fulfill quasi static modeling assumptions.

Additional modeling assumptions and limitations of the library are presented. The compatibility of the quasi static with the transient fundamental wave machine library is discussed. Simulation examples of quasi static and transient machine models are compared.

The new package opens a new field of applications for mobility and industry applications since the quasi static machine models have a very high simulation performance. Possible application examples are presented and didactic aspects of the library are discussed.

References

- [1] C. Kral and A. Haumer, “The new fundamental wave library for modeling rotating electrical three phase machines,” *8th International Modelica Conference*, 2011.
- [2] A. Haumer, C. Kral, J. V. Gragger, and H. Kapeller, “Quasi-stationary modeling and simulation of electrical circuits using complex phasors,” *International Modelica Conference, 6th, Bielefeld, Germany*, pp. 229–236, 2008.
- [3] N. Rabe, “An approach for modelling quasi-stationary magnetic circuits,” *9th International Modelica Conference*, 2012.
- [4] C. Kral, A. Haumer, and S. B. Lee, “A practical thermal model for the estimation of permanent magnet and stator winding temperatures,” *IEEE Transactions on Power Electronics*, vol. 29, no. 1, pp. 455–464, 2014.
- [5] A. Iqbal, E. Levi, M. Jones, and S. Vukosavic, “Generalised sinusoidal PWM with harmonic injection for multi-phase VSIs,” *PESC '06. 37th IEEE Power Electronics Specialists Conference*, 2006, 2006.
- [6] M. J. Duran, F. Barrero, and S. Toral, “Multi-phase space vector pulse width modulation: Applications and strategies,” *International Conference on Renewable Energies and Power Quality, ICREPQ 2007*, 2007.
- [7] D. Dujic, E. Levi, M. Jones, G. Grandi, G. Serra, and A. Tani, “Continuous PWM techniques for sinusoidal voltage generation with seven-phase voltage source inverters,” *Power Electronics Specialists Conference, 2007. PESC 2007. IEEE*, pp. 47–52, 2007.
- [8] S. Halasz, “PWM strategies of multi-phase inverters,” *IECON 2008. 34th Annual Conference of IEEE Industrial Electronics*, 2008.
- [9] D. Dujic, M. Jones, and E. Levi, “Generalised space vector PWM for sinusoidal output voltage generation with multiphase voltage source inverters,” *International Journal of Industrial Electronics and Drives*, vol. 1, no. 1, 2009.

- [10] M. Mengoni, *Modulation Techniques for Multi-Phase Converters and Control Strategies for Multi-Phase Electric Drives*. PhD thesis, University of Bologna, 2010.